PharmaSUG China 2024 - Paper DV-10054

# A Story-telling Patient Profile One-page Graph Using GTL

Huizheng Yang, Pfizer (China) Research and Development Co.,Ltd. China
Qili Fei, Pfizer (China) Research and Development Co.,Ltd. China

## ABSTRACT

When generating reports for a clinical trial, the reviewers may request patient profile individual plots for the patients they are interested in. How to improve interactive data visualization of all the interested data? In this paper, we will introduce a patient profile plot that combines IP exposure, concomitant medications, dose-normalized PK trough concentrations and liver function lab tests in a one-page graph using the SAS ® Graph Template Language (GTL) to build an informative and meaningful data story. The application of this graph can be expanded to other domains to enhance reviewer's understanding of the safety performance and risks.

## INTRODUCTION

The SAS ® Graph Template Language (GTL) procedure is a powerful tool to create complicated graphs based on CDSIC domains. It can use multiple plot patterns in one template. i.e., one graph can display both scatter plot and line plot.

Patient profile usually includes listings and plots for individual subjects. The plots are often present separately based on safety domains over a timeline. Adverse Events, Concomitant Medications, dosage and other information are usually displayed in separate graphs. The Y-axis and X-axis may vary across the domains and hard to find the interaction visually.

This paper will show you a way to present the safety data including the lab liver function test results, IP exposure, concomitant medications, PK concentrations into one graph using GTL.

## PATIENT PROFILE GRAPH EXAMPLE




**Figure 1. Plot Including All Data Parts versus Study Day**

We used the simulated data for the presentation. This example plot (Figure 1) aims to explore the safety data interactions based on study days for the subjects with suspected DILI. The plot combines lab liver function test results, IP exposure, concomitant medications, PK dose-normalized trough concentrations of each subject into one-page graph. As the most interested data, the liver function lab tests are displayed in the main lower part of the graph. The X axes are study days. The Y axes on the left are the multiple ULNs.

Drug A dose-normalized trough concentration data is presented in the same position with the liver function lab tests to compare the trend between the lab and PK data. The Y axes on the right is dose-normalized trough concentration. The brown bold line is set to distinguish from the lab test.

Concomitant medications are presented in the horizontal blue line based on the study days.

The dose exposure is showed at the top of the graph with the pink bar for the Drug A, and dot plot for the Drug B under the Drug A bar.

## DATA SET PREPARATION AND GTL PROCESS

### LAB LIVER FUNCTION TESTS (LFT)

For the lab LFT results, the plot needs to be processed into two parts: 1) within or before the exposure: line plot with symbol circle; 2) after the exposure: line plot with symbol X. To show the line plot with different symbols, we prepared three data sets for the three segments of the line plot: 1) values before or during the exposure, 2) values after the exposure, 3) the last value during the exposure and the first value after the exposure.

```
data adlb_liver_selected;
   set adlb_liver;
   where usubjid="&&usubjid&i.";
   AVAL_ULN=AVAL/ANRHI;
   if PARAMCD = 'ALT' then _PARAMN=1;
   if PARAMCD = 'AST' then _PARAMN=2;
   if PARAMCD = 'ALP' then _PARAMN=3;
   if PARAMCD = 'TB' then _PARAMN=4;
   keep USUBJID PARAMCD _PARAMN ADT ADTM ADY AVAL ANRHI AVAL_ULN TRTEDT;
run;

proc format;
   value _PARAM
   1 = "ALT/ULN"
   2 = "AST/ULN"
   3 = "ALP/ULN"
   4 = "TB/ULN"
   ;
run;

/* Prepare the data sets for the values before or during the exposure */
data adlb_liver_final1;
   set adlb_liver_selected;
   _x_lb1=ADY; _y_lb1=AVAL_ULN;
   _cat="adlb";
   where ADT<=TRTEDT;
   format _PARAMN _PARAM.;
   keep USUBJID _PARAMN _x_lb1 _y_lb1 _cat;
   proc sort; by USUBJID _PARAMN _x_lb1;
run;
```

```
/* Prepare the data sets for the values after the exposure */
data adlb_liver_final2;
   set adlb_liver_selected;
   _x_lb2=ADY; _y_lb2=AVAL_ULN;
   _cat="adlb";
   where ADT>TRTEDT;
   format _PARAMN _PARAM.;
   keep USUBJID _PARAMN _x_lb2 _y_lb2 _cat;
   proc sort; by USUBJID _PARAMN _x_lb2;
run;

/* Third data set: the last value during the exposure and the first value
after the exposure to connect the two series plot */
data adlb_liver_final3_1;
   set adlb_liver_final1;
   by USUBJID _PARAMN _x_lb1;
   _x_lb3=_x_lb1; _y_lb3=_y_lb1;
   if last._PARAMN;
   keep USUBJID _PARAMN _x_lb3 _y_lb3 _cat;
run;

data adlb_liver_final3_2;
   set adlb_liver_final2;
   by USUBJID _PARAMN _x_lb2;
   _x_lb3=_x_lb2; _y_lb3=_y_lb2;
   if first._PARAMN;
   keep USUBJID _PARAMN _x_lb3 _y_lb3 _cat;
run;

data adlb_liver_final3;
   set adlb_liver_final3_1 adlb_liver_final3_2;
   proc sort; by USUBJID _PARAMN _x_lb3;
run;
```

In the GTL procedure, we need to apply different peak ULNs and marks on Y-axis based on the different individual peak ALT ULNs.

| When Peak ALT is | Present Y-axis with Y- axis up to | Marks on Y-axis (x ULN) |
|---|---|---|
| <= 5x ULN | 6x ULN | 1, 3, 5 |
| <= 10x ULN | 11x ULN | 1, 3, 5, 8, 10 |
| <= 20x ULN | 21x ULN | 1, 3, 5, 8, 10, 15, 20 |
| <= 35x ULN | 35x ULN | 1, 3, 5, 8, 10, 15, 20, 30, 35 |

**Table 1. Requirements of Y-axis with Different Peak ALTs**

We set the detailed options in GTL and apply the template in Proc SGRENDER to create the LFT part plot (Figure 2).

```
proc template;
   define statgraph plot&i.;
   dynamic etitle1 etitle2;
   begingraph / designwidth=9in designheight=6in;
```

```
  entrytitle textattrs=(weight=bold size=8pt) etitle1;
  entrytitle textattrs=(weight=bold size=8pt) etitle2;
  layout overlay/

  /* Using study day for the X-axis */
  xaxisopts=(offsetmin=0.02 offsetmax=0.02 griddisplay=off label="Study
  Day" labelattrs=(size=7pt) tickvalueattrs=(size=5pt)
  linearopts=(viewmin=-1 viewmax=100 tickvaluefitpolicy=none
  tickvaluelist=(-1 1 10 20 30 40 50 60 70 80 90 100)))

  /* Using multiple ULN for the Y-axis. The max value depends on the Peak
  ALT ULNs. Offsetmax is adjusted to expand upper area of the plot */
  yaxisopts=(offsetmin=0.02 offsetmax=0.4 griddisplay=off label="x ULN"
  labelattrs=(size=7pt) tickvalueattrs=(size=5pt) linearopts=(viewmin=0
  viewmax=&&lbtickmax&i. tickvaluefitpolicy=none
  tickvaluelist=(&&lbytick&i.)));

  /* Series plot before or during the exposure */
  seriesplot x=_x_lb1 y=_y_lb1/ group=_PARAMN includemissinggroup=false
  name="param" display=all markerattrs=(size=4pt);
  discretelegend "param"/displayclipped=true border=true across=4
  halign=left valign=bottom;

  /* Series plot after the exposure */
  seriesplot x=_x_lb2 y=_y_lb2/ group=_PARAMN includemissinggroup=false
  display=all markerattrs=(symbol=X size=4pt);

  /* The last value during the exposure and the first value after the
  exposure to connect the two series plot */
  seriesplot x=_x_lb3 y=_y_lb3/ group=_PARAMN includemissinggroup=false;

  /* Reference line for the Y-axis of the different peak ALT ULNs */
  referenceline y=1/lineattrs=(pattern=34);
  referenceline y=3/lineattrs=(pattern=34);
  referenceline y=5/lineattrs=(pattern=34);
  referenceline y=8/lineattrs=(pattern=34);
  referenceline y=10/lineattrs=(pattern=34);
  %if %index(&&lbytick&i., 15) %then %do; referenceline
  y=15/lineattrs=(pattern=34); %end;
  %if %index(&&lbytick&i., 20) %then %do; referenceline
  y=20/lineattrs=(pattern=34); %end;
  %if %index(&&lbytick&i., 30) %then %do; referenceline
  y=30/lineattrs=(pattern=34); %end;
  %if %index(&&lbytick&i., 35) %then %do; referenceline
  y=35/lineattrs=(pattern=34); %end;
  endlayout;
  endgraph;
end;
```




**Figure 2. LFT Plot versus Study Day**

## EXPOSURE

The example dose information here is an open label study with two drugs: Drug A is continuous dosed from day 7 to day 43. Drug B doses on three separate days. Drug A dosed from day 7 to day 43 with escalation of dosage: 20mg, 40mg, 80mg, 120mg, 160mg, 200mg, 220mg and 240mg. Drug B dosed on day 1, 15 and 41.

| Row | USUBJID | EXTRT | EXDOSE | EXDOSU | EXDOSFRQ | EXSTDY | EXENDY |
|---|---|---|---|---|---|---|---|
| 1 | 01-1001-10011001 | Drug B | 150 | mg | ONCE | 1 | 1 |
| 2 | 01-1001-10011001 | Drug B | 150 | mg | ONCE | 15 | 15 |
| 3 | 01-1001-10011001 | Drug B | 150 | mg | ONCE | 41 | 41 |
| 4 | 01-1001-10011001 | Drug A | 20 | mg | QD | 7 | 10 |
| 5 | 01-1001-10011001 | Drug A | 40 | mg | QD | 11 | 14 |
| 6 | 01-1001-10011001 | Drug A | 80 | mg | QD | 15 | 23 |
| 7 | 01-1001-10011001 | Drug A | 120 | mg | QD | 24 | 27 |
| 8 | 01-1001-10011001 | Drug A | 160 | mg | QD | 28 | 31 |
| 9 | 01-1001-10011001 | Drug A | 200 | mg | QD | 32 | 35 |
| 10 | 01-1001-10011001 | Drug A | 220 | mg | QD | 36 | 39 |
| 11 | 01-1001-10011001 | Drug A | 240 | mg | QD | 40 | 43 |

**Table 2. Exposure Data**

Drug A displays as pink bar on the top the plot. Drug B displays as dot plot under the Drug A. We need to process the data of two drugs separately. If the Drug A interprets during the study, we need to separate the exposure into multiple parts. The vertical reference lines are created as macro variables to indicate the start dose day, dosage and end of administration information.

```
data ex;
  set dataprot.ex;
  where usubjid="&&usubjid&i.";
```

```
  proc sort; by USUBJID;
run;

data ex_selected;
  set ex;
  keep USUBJID EXTRT EXDOSE EXDOSU EXDOSFRQ EXSTDTC EXENDTC EXSTDY EXENDY;
  proc sort; by USUBJID EXTRT EXSTDY EXENDY;
run;

data ex_selected;
  set ex_selected;
  by USUBJID EXTRT EXSTDY EXENDY;
  length _EXTRT $100;

  /* Create _EXTRT to combined the Drug Name, Dosage and Dose unit together
  for the plot label */
  _EXTRT=catx(", ", EXTRT, catx(" ", put(EXDOSE,best.),EXDOSU));
  if EXTRT="Drug A" then _EXTRT=strip(_EXTRT)|| " "|| strip(EXDOSFRQ);
run;

/* Data set of Drug B */
data ex_final_pre1;
  set ex_selected;
  where EXTRT in ("Drug B");
  minEXDY=EXSTDY; maxEXDY=EXENDY;
  keep USUBJID EXTRT _EXTRT minEXDY maxEXDY;
run;

/* Data set of Drug A */
data ex_pre2;
  set ex_selected;
  where EXTRT in ("Drug A");
  proc sort; by USUBJID EXSTDY EXENDY;
run;

data ex_pre2;
  set ex_pre2;
  by USUBJID EXSTDY EXENDY;
  /* Separate parts for Drug A if there is dose interruption */
  retain part;
  if first.USUBJID then part=0;
    if EXSTDY ne lag(EXENDY)+1 then do;
    part=part+1;
  end;
run;

proc sql noprint;
  create table ex_final_pre2 as
  select distinct USUBJID, EXTRT, part, min(EXSTDY) as minEXDY, max(EXENDY)
  as maxEXDY
  from ex_pre2
  group by USUBJID, EXTRT, part;
quit;

data ex_final;
  set ex_final_pre1 ex_final_pre2;
  if EXTRT eq "Drug B" then do;
```

```
      _x_trt1=minEXDY; _EXTRTlabel=_EXTRT;

      /*Set the same position for the Drug B across subjects: Using the max
      Y-axis * 0.48 to fix the position of Drug B plot for the y-axis */
      _y_trt1=&&lbtickmax&i. + &&lbtickmax&i. * 0.48;
      output;
    end;
    if EXTRT eq "Drug A" then do;
      _x_trt2_st=minEXDY; _x_trt2_en=maxEXDY;

      /* Set the same position for the Drug A across subjects: Using the max
      Y-axis * 0.6 to fix the position of Drug A plot for the y-axis */
      _y_trt2=&&lbtickmax&i. + &&lbtickmax&i. * 0.6;
      output;
    end;
  run;

  /* Create macro variables for reference line on x-axis */
  proc sql noprint;
     create table ex_pf_st as
     select distinct USUBJID, _EXTRT, min(EXSTDY) as minEXDY, max(EXENDY) as
  maxEXDY
     from ex_selected where EXTRT="Drug A"
     group by USUBJID, _EXTRT;
  quit;

  proc sort data=ex_pf_st; by usubjid minEXDY; run;

  data ex_pf_st;
     set ex_pf_st end=eof;
     by usubjid minEXDY;
     if first.minEXDY then do;
           pfdosen + 1;
           call symput ('pfdose'||compress(put(pfdosen,3.0)),cats(_EXTRT));
           call symput
  ('pfstdy'||compress(put(pfdosen,3.0)),strip(put(minEXDY,best.)));
      end;
      if eof then do;
           call symput ('pfdosen',strip(put(pfdosen,best.)));
           call symput ('pflaendy',strip(put(maxEXDY,best.)));
      end;
  run;

  %let pfdosen&i.=&pfdosen.;
  %let pflaendy&i.=&pflaendy.;
  %do b=1 %to &pfdosen.;
     %let pfdose&i._&b.=&&pfdose&b.;
     %let pfstdy&i._&b.=&&pfstdy&b.;
  %end;
```

In the GTL procedure, we called macro variable &&pfstdy&i for the reference lines, &&pfdose&i._&k. for the label of the reference lines and &&pflaendy&i for the “last administration” reference lines. Pink bar plot displays drug A dose information based on study days. Dot plot shows the single dose drug B under drug A.

```
  /* Dots on the plot for Drug B dosed on day 1, 15 and 41 */
```

```
scatterplot x=_x_trt1 y=_y_trt1/ datalabel=_EXTRTlabel
datalabelattrs=(size=4pt) datalabelposition=top   markerattrs=(color=grey
symbol=circlefilled size=5pt);

/* Pink bar on the plot for Drug A dosed from day 7 to day 43 with
escalation of dosage: 20mg, 40mg, 80mg, 120mg, 160mg, 200mg, 220mg and
240mg */
highlowplot y=_y_trt2 low=_x_trt2_st high=_x_trt2_en/ display=(fill)
type=bar barwidth=0.2 fillattrs=(color=pink);

/* Vertical reference lines for the Drug A dosage */
%do k=1 %to &&pfdosen&i.;
referenceline x=&&pfstdy&i._&k./ lineattrs=(pattern=34)
curvelabel="&&pfdose&i._&k." curvelabelattrs=(size=5pt);
%end;
referenceline x=&&pflaendy&i./ lineattrs=(pattern=34) curvelabel="Last
Administration" curvelabelattrs=(size=5pt);
referenceline x=1/lineattrs=(pattern=34) curvelabel="Treatment Start"
curvelabelattrs=(size=5pt);
```

Figure 3 shows the graphs of exposure with different dosing situations of Drug A. The left plot shows the subject doses drug A without interruption till the end of administration. The right plot shows the subject doses with interruption on Day 20 and Day 30. The plot shows the pink bar plot starts on day 1 and ends on day 19, starts on day 21 and ends on day 29, starts on day 31 and ends on day 43.



**Figure 3. Exposure Plot with Different Dosing Situations of Drug A versus Study Day**

## CONCOMITANT MEDICATIONS

The concomitant medications are presented under the IP exposure in the graph. When the concomitant medication is ongoing, we need show arrow on the plot until the treatment end day.
We have three types of CM data and need to treat in separate ways: 1) start and end on the same day, 2) start and end on different days, 3) ongoing till the last administration.

```
data adcm;
   set datvprot.adcm;
   where usubjid="&&usubjid&i.";
   if ONPERFL="Y" and not missing(CMDECOD);
   if missing(AENDY) and not missing(TRTEDT) then AENDY=TRTEDT-TRTSDT+1;
   proc sort; by USUBJID;
run;
```

```
data adcm_selected;
   set adcm;
   keep USUBJID CMDECOD CMSTDTC CMENDTC ASTDY AENDY;
   proc sort nodupkey; by USUBJID CMDECOD ASTDY AENDY;
run;

proc sort data=adcm_selected out=cmdecod_selected(keep=USUBJID CMDECOD)
nodupkey;
   by USUBJID CMDECOD;
run;

/* Using the max Y-axis * proper ratios to fix the position of different
CMDECOD for the y-axis */
data cmdecod_selected;
    set cmdecod_selected;
    by USUBJID CMDECOD;
    retain _y_cm;
    if first.USUBJID then _y_cm=&&lbtickmax&i. + &&lbtickmax&i. * 0.4;
    _y_cm = _y_cm - &&lbtickmax&i. * 0.08;
run;

/* Create macro variables for CMDECOD which will be used in GTL */
data cmdecod_selected;
   set cmdecod_selected end=eof;
   by USUBJID CMDECOD;
   if first.CMDECOD then do;
     _cmcode + 1;
     call symput ('CMDECOD'||compress(put(_cmcode,3.0)),cats(CMDECOD));
   end;
   if eof then call symput ('ncmdecod',strip(put(_cmcode,best.)));
run;

%let ncmdecod&i.=&ncmdecod.;
%do a=1 %to &ncmdecod.;
   %let CMDECOD&i._&a.=&&CMDECOD&a.;
%end;

data adcm_selected;
   merge adcm_selected(in=a) cmdecod_selected;
   by USUBJID CMDECOD;
   if a;
run;

data adcm_final(keep=USUBJID _x_cm _y_cm _xcmong CMDECOD _cmcode ASTDY
AENDY _cat);
   set adcm_selected;
   _cat="adcm";

   /* CM started and ended on the same day */
   if not missing(ASTDY) and ASTDY=AENDY then do;
     _x_cm=ASTDY;
     output;
   end;

   /* CM started and ended on the different day */
   else if not missing(ASTDY) and ASTDY ^= AENDY then do;
     _x_cm=ASTDY;
```

```
      output;
      _x_cm=AENDY;
      output;
    end;

    /* CM started and ongoing during the treatment */
    if missing(CMENDTC) then do;
      _xcmong=AENDY;
      output;
    end;
  run;

  proc sort data=adcm_final; by USUBJID CMDECOD ASTDY AENDY; run;

  data adcm_final;
  set adcm_final;
    by USUBJID CMDECOD ASTDY AENDY;
    /* When the CM occurs many times in one subject, we need only one label
    for the same concomitant medications */
    if last.CMDECOD then cmdatalabel=strip(CMDECOD);
  run;
```

In the GTL procedure, we use series plot for the three different situations. Additionally, the scatter plot is added for the ongoing CM data. The TriangleRightFilled symbol is set as arrow to cover the end dot of ongoing CM. To show the CM records start and end on the same day as a dot, we use the option display=all.

```
  %do j=1 %to &&ncmdecod&i.;
    /* show the CM records start and end on the same day as a dot */
    seriesplot x=_x_cm y=_y_cm/ group=astdy display=all
    curvelabel=cmdatalabel curvelabelposition=end curvelabelattrs=(size=5pt
    color=blue) lineattrs=(pattern=solid thickness=1pt color=blue)
    markerattrs=(size=3pt symbol=circlefilled color=blue);
  %end;

  /* Scatter plot to display the CM ongoing */
  scatterplot x=_xcmong y=_y_cm/ markerattrs=(size=6pt
  symbol=TriangleRightFilled color=blue);
```




**Figure 4. Plot of Concomitant Medications versus Study Day**

## PK DOSE-NORMALIZED TROUGH CONCENTRATION

The PK dose-normalized trough concentration is presented on the lower part of the plot. Same as the LFT plots, the data is processed into three parts: 1) values during the exposure, 2) values after the exposure, 3) the last value during the exposure and the first value after the exposure.

```
data adpc_selected;
   set datvprot.adpc;
   where usubjid="&&usubjid&i.";
   if PARAM="Drug A Dose-normalized Trough Concentration";
run;

/* Prepare the data sets for the values during the exposure */
data adpc_final1;
   set adpc_selected;
   _x_pc1=ADY; _y_pc1=AVAL;
   _cat="adpc";
   where ADT<=TRTEDT;
   keep USUBJID PARAM _x_pc1 _y_pc1 _cat;
   proc sort; by USUBJID PARAM _x_pc1;
run;

/* Prepare the data sets for the values after the exposure */
data adpc_final2;
   set adpc_selected;
   _x_pc2=ADY; _y_pc2=AVAL;
   _cat="adpc";
   where ADT>TRTEDT;
   keep USUBJID PARAM _x_pc2 _y_pc2 _cat;
   proc sort; by USUBJID PARAM _x_pc2;
run;

/* The last value during the exposure and the first value after the
exposure to connect the two series plot */
```

```
data adpc_final3_1;
  set adpc_final1;
  by USUBJID PARAM _x_pc1;
  _x_pc3=_x_pc1; _y_pc3=_y_pc1;
  if last.PARAM;
  keep USUBJID PARAM _x_pc3 _y_pc3 _cat;
run;

data adpc_final3_2;
  set adpc_final2;
  by USUBJID PARAM _x_pc2;
  _x_pc3=_x_pc2; _y_pc3=_y_pc2;
  if first.PARAM;
  keep USUBJID PARAM _x_pc3 _y_pc3 _cat;
run;

data adpc_final;
  set adpc_final1 adpc_final2 adpc_final3_1 adpc_final3_2;
run;
```

In the GTL procedure, same as the LFT plot, we combined three parts of the PK data into one brown bold line. The right Y-axis is set for the PK dose-normalized trough concentration values.

```
/* The second Y-axis for the PK dose-normalized trough concentration, same
offsetmax as the first Y-axis is set to expand upper area of the plot */
y2axisopts=(offsetmin=0.02 offsetmax=0.4 griddisplay=off label="Drug A
Dose-normalized Trough Concentration (ng/mL/mg)" labelattrs=(size=7pt)
tickvalueattrs=(size=5pt) linearopts=(viewmin=0 viewmax=500
tickvaluefitpolicy=none tickvaluelist=(0 100 200 300 400 500)));

/* Series plot during the exposure */
seriesplot x=_x_pc1 y=_y_pc1/ yaxis=y2 group=param
includemissinggroup=false name="pcparam" display=all
lineattrs=(pattern=solid thickness=2.4pt color=brown)
markerattrs=(color=brown symbol=circlefilled size=5.5pt);
discretelegend "pcparam"/displayclipped=true border=true halign=right
valign=bottom;

/* Series plot after the exposure */
seriesplot x=_x_pc2 y=_y_pc2/ yaxis=y2 group=param
includemissinggroup=false display=all lineattrs=(pattern=solid
thickness=2.4pt color=brown) markerattrs=(color=brown symbol=X size=5.5pt);

/* The last value during the exposure and the first value after the
exposure to connect the two series plot */
seriesplot x=_x_pc3 y=_y_pc3/ yaxis=y2 group=param
includemissinggroup=false lineattrs=(pattern=solid thickness=2.4pt
color=brown);
```




**Figure 5. Plot of Trough Concentration versus Study Day**

## ONE-PAGE GRAPH

To combine all parts into one graph, we set all prepared data sets into one data set and set the options for each part within a single GTL. Finally, we used a single sgrender procedure to generate the one-page graph (Figure 1).
In legend box, we set option includemissinggroup=false in the GTL to avoid showing empty groups in the final reporting data.
We also set a proper offsetmax in yaxisopts and y2axisopts to expand the upper area of the plot to display the plots of exposure and concomitant medications. Offsetmin/offsetmax of yaxisopts and y2axisopts are same to make the minimum and maximum ticks of left and right Y-axis be at the same horizontal level.

```
data final&i.;
   set adlb_liver_final1 adlb_liver_final2 adlb_liver_final3 adcm_final
   ex_final adpc_final;
run;

proc template;
   define statgraph plot&i.;
   dynamic etitle1 etitle2;
   begingraph / designwidth=9in designheight=6in;
   entrytitle textattrs=(weight=bold size=8pt) etitle1;
   entrytitle textattrs=(weight=bold size=8pt) etitle2;
   layout overlay/
   xaxisopts=(offsetmin=0.02 offsetmax=0.02 griddisplay=off label="Study
   Day" labelattrs=(size=7pt) tickvalueattrs=(size=5pt)
   linearopts=(viewmin=-1 viewmax=100 tickvaluefitpolicy=none
   tickvaluelist=(-1 1 10 20 30 40 50 60 70 80 90 100)))
   yaxisopts=(offsetmin=0.02 offsetmax=0.4 griddisplay=off label="x ULN"
   labelattrs=(size=7pt) tickvalueattrs=(size=5pt) linearopts=(viewmin=0
   viewmax=&&lbtickmax&i. tickvaluefitpolicy=none
   tickvaluelist=(&&lbytick&i.)))
   y2axisopts=(offsetmin=0.02 offsetmax=0.4 griddisplay=off label="Drug A
   Dose-normalized Trough Concentration (ng/mL/mg)" labelattrs=(size=7pt)
   tickvalueattrs=(size=5pt) linearopts=(viewmin=0 viewmax=500
   tickvaluefitpolicy=none tickvaluelist=(0 100 200 300 400 500)));

   …

   endlayout;
```

```
    endgraph;
  end;

  proc sgrender data= final&i. template=plot&i.;
    dynamic etitle1="Liver Biochemistry (Ratio to ULN), IP Exposure,
    Concomitant Medication and Drug A Dose-normalized Trough Concentration";
    dynamic etitle2="Unique Participant ID: &&usubjid&i.";
  run;
```

## CONCLUSION

Combined safety information into one-page graph gives the reviewer a whole image of the patient safety performance. The relevant data presented in one graph in the suitable pattern showed the interactions intuitively. When preparing the safety reporting data, all the situations should be considered for the following GTL procedure. Different plots could be used for one domain to optimize the visualization.

## REFERENCES

SAS GTL User's Guide
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatug/titlepage.htm

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Huizheng Yang
Enterprise: Pfizer (China) Research and Development Co.,Ltd. China
E-mail: Huizheng.Yang@pfizer.com

Name: Qili Fei
Enterprise: Pfizer (China) Research and Development Co.,Ltd. China
E-mail: Qili.Fei@pfizer.com